

# Control automático del nivel de luz.

Dose, Aldo – Franke, Favián – Lima, Carlos

[aldo\\_dose@yahoo.com.ar](mailto:aldo_dose@yahoo.com.ar) – [favian.franke@gmail.com](mailto:favian.franke@gmail.com) – [carlos\\_ariel20081@hotmail.com](mailto:carlos_ariel20081@hotmail.com)

## **Eje temático:**

4. Aplicaciones en dispositivos hogareños

## **Palabras claves:**

DIMMER – CRUCE POR CERO – ÁNGULO DE DISPARO – LDR – TRIAC – MICROCONTROLADOR.

## **Resumen**

Inicialmente el proyecto fue desarrollado como una tarea que se nos asignó en la cátedra Técnicas Digitales 2, dictada en la carrera Ingeniería Electrónica de la Facultad de Ingeniería de la UNaM. Para la presentación del proyecto en el congreso, se han tomado una serie de decisiones respecto al diseño del hardware, realizando varias modificaciones.

El desarrollo presentado en esta ponencia, consiste en un sistema, que por medio del sensado del nivel de luminancia y el control del ángulo de disparo de un triac, logra controlar el nivel de luz, variando la potencia lumínica de una lámpara incandescente. El nivel de luz deseado por el usuario, puede ser seteado por medio de un potenciómetro de manera gradual.

La implementación del proyecto se realizó con un microcontrolador de la firma Freescale Semiconductor, el MC68HC908QT4, el cual, por medio del cálculo de la ley de control logra el objetivo deseado.

La etapa de potencia fue diseñada para poder manejar una potencia máxima de hasta 1.200 Watts.

Se trató, y con buenos resultados, lograr la minimización de componentes utilizados y el costo de realización del proyecto.

## Introducción

Los dimmer son dispositivos usados para regular el voltaje de una o varias lámparas. Así, es posible variar la intensidad de la luz, siempre y cuando las propiedades de la lámpara lo permitan, como es el caso de las lámparas incandescentes, como la aquí utilizada.

La disminución del valor eficaz en la lámpara se logra “desconectándola” de la red por un intervalo de tiempo que va desde el cruce por cero de la tensión alterna, hasta el momento deseado. Esto lo logramos actuando sobre la señal de cebado del triac. Para dispararlo se optó por el control por pulsos, ya que éste trae las siguientes ventajas:

- Permite una potencia de pico superior a la potencia media de la puerta admisible.
- Pueden aplicarse criterios de tolerancia más amplios al circuito de disparo.
- Es posible reducir a un valor mínimo el retardo entre la señal de puerta y la subida de la corriente de ánodo. Lo que permite obtener una sincronización muy precisa.
- Se reduce la potencia que el circuito de mando invierte para el disparo del dispositivo.

Para sensar el nivel de luz se utilizó una fotorresistencia, la cual es un componente electrónico cuya resistencia disminuye con el aumento de intensidad de luz incidente. Puede también ser llamado fotorresistor, fotoconductor, célula fotoeléctrica o resistor dependiente de la luz, cuyas siglas, LDR, se originan de su nombre en inglés *light-dependent resistor*.

El valor de resistencia eléctrica de un LDR es bajo cuando hay luz incidiendo en él (puede descender hasta 50 ohms) y muy alto cuando está a oscuras (varios megaohms). De esta característica de funcionamiento nos valimos para, por medio de un divisor resistivo, obtener la lectura.

## Hardware

Siguiendo la premisa de minimizar los costos de realización el resultado que obtuvimos a la hora del diseño del hardware fue:

### Fuente de alimentación

Para el proyecto de la fuente se optó por una fuente sin transformador, la que se vale de diodos zeners para brindar la tensión de alimentación regulada a los distintos componentes del circuito. El diodo zener *D1* limita la carga del capacitor *C2* a un valor máximo de 12 V, mientras que el diodo zener *D2*, regula con *R3* la tensión a 5.1 V. La fuente se conecta directamente a la red, gracias a que la red *R1*, *C1* está proyectada para reducir la tensión lo suficiente para no dañar a los restantes componentes del circuito.

### Microprocesador

La búsqueda del microcontrolador que mejor se ajuste a nuestras necesidades la basamos en que este posea 2 canales de entradas analógicas (uno para la referencia, y otro para el sensado de la tensión sobre el LDR), un pin de entrada para detección de la señal de sincronismo (cruce por cero de la tensión de red), un pin de salida para el control de la puerta del triac, un modulo que permita la temporización.

Luego de buscar en las características de los distintos microcontroladores disponibles en el mercado, nos decidimos por el MC68HC908QT4, microcontrolador de la firma Freescale Semiconductor.

### Señal de referencia

Para generar la señal de referencia se utilizó un potenciómetro como divisor resistivo.

### Sensor de intensidad luminosa


Aprovechando que la fotorresistencia varia su impedancia con la luz, se la utilizó con una resistencia fija para formar un divisor resistivo básico.

## Generación del sincronismo

El circuito de sincronismo nos posibilita "saber" cuando la tensión pasa por cero, por lo que a veces también se la denomina detector de cruce por cero. A partir del instante en que se produce el cruce por cero, se inicia la temporización para retardar el disparo del triac según corresponda.

## Circuito de disparo del triac

Como ya se dijo, el disparo del triac se realiza por pulsos, los cuales los genera el microprocesador. Como se puede ver en el circuito, se conectó el pin de salida a la base del transistor Q1, el cual se encarga de disparar el triac



## Software

Lo que realiza el programa es la inicialización de variables y de puertos (configuración de los conversores y de los módulos de temporización). Luego de esta configuración se procede a la captura del primer Set-Point y la salida, valores con los que se determina el error, y con este el tiempo necesario a esperar para realizar el disparo del triac (acción de control del sistema), como el error puede ser positivo o negativo se trabaja con siete bits, el ultimo (octavo bit) nos da el signo, para ello se realiza un

desplazamiento a la derecha. Luego se habilita la interrupción por IRQ (Pin 5).

El sensado del nivel de iluminación y el cálculo de la acción de control se realizan cada ciclo de la red, es decir cada vez que se produce la interrupción debida al cruce por cero (señal de sincronismo). Dicha ley de control se basa en calcular en error entre la referencia y la señal sensada, luego se calcula en retardo actual, que es igual al retardo anterior menos el error.


La interrupción que provoca el llamado a esta subrutina se realiza a

través del pin IRQ, activada por la señal de sincronismo. El seteo del Output-compare se realiza para disparar el triac por medio del pin de salida (Pin 7).


Los valores de retardos son limitados tanto en el limite superior como en el

inferior, para evitar que se produzca un desborde del contador (lo que acarrearía un retardo incorrecto, mayor a 10 mseg, o para evitar un retardo tan chico que en la siguiente transición produzca un mal calculo del retardo).


### El diagrama de flujo del programa




## Resultados experimentales



Señales de sincronismo con la red y pulsos de disparo aplicados a la puerta del triac para distintos tiempos de retardo.



Corriente que circula por el foco para distintos ángulos de disparo.



Respuesta a la referencia en escalón, se aprecia la referencia y el nivel de luz logrado.

## **Conclusiones**

Como conclusión de la del trabajo realizado, podemos decir que se cumplió con el objetivo plantea en un comienzo, se regulo la "luz" en un punto determinado. Podemos agregar además, que el objetivo se cumplió, como era una meta también, minimizando la inversión de materiales.

Como punto deficiente, podemos resaltar que el comportamiento de la planta ante una referencia en escalón presenta un sobre impulso, el cual no es elevado y no se aprecia en el brillo del foco, pero se podría llegar a disminuirlo si se utiliza una ley de control un poco más elaborada.

## Código fuente

```
PORTA      EQU $0000      ; Direccion del puerto A
DDRA       EQU $0004
INTSCR     EQU $001D      ; Registro de estado y control del IRQ
CONFIG2    EQU $001E
CONFIG1    EQU $001F
TSC        EQU $0020
TSC1       EQU $0028      ; Canal 1 del modulo TIM
TCH1H      EQU $0029
TCH1L      EQU $002A
ADSCR      EQU $003C      ; Registro de estado y control del ADC
ADR        EQU $003E      ; Donde se guarda el resultado de la conv.
           ; Definicion de los bits para configuracion
           ; Config 1
COPD       EQU 0          ; Permite activar/desactivar el COP
           ; Puerto A
PTA0       EQU 0
           ; Direcciones
RAM         EQU $0080      ; Inicio de la RAM
FLASH      EQU $EE00      ; Inicio de la FLASH
RESET      EQU $FFFE
           ; Configuracion del ADC
ADC0       EQU 5          ; Para cambiar entre conv continua y una sola
           ; CONFIGURACION PARA EL LDR (PTA5) Pin2
ADCLDR     EQU %00000011  ; Final de conversion (no se escribe)
           ; Deshabilita la interrupcion de final de conv
           ; Una conversion
           ; El resto eleccion del canal ADC (Canal 3)
           ; CONFIGURACION PARA LA REFERENCIA (PTA4) Pin3
ADC_SP     EQU %00000010  ; Final de conversion (no se escribe)
           ; Deshabilita la interrupcion de final de conv
           ; Una conversion
           ; El resto eleccion del canal ADC (Canal 2)
           ; Modulo TIM
TRST       EQU 4          ; Para poner el TIM counter a 0
CHIE       EQU 6          ; Para habilitar las interrupciones de del TIM
TSTOP      EQU 5          ; Para detener (1) /iniciar (0) el contador
CH_F       EQU 7          ; Flag
OUTCOM     EQU %01010000  ; Se configura el output-compare
           ; Interrupcion habilitada
           ; Software compare only

           ORG   RAM
RETANTH    ds    1        ; Tiempo anterior TRIAC apagado
RETANTL    ds    1        ; Tiempo anterior TRIAC apagado
RETACTH    ds    1        ; Tiempo actual TRIAC apagado
REACTL     ds    1        ; Tiempo actual TRIAC apagado
RETMAXH    ds    1
RETMAXL    ds    1
REF        ds    1
SENSOR     ds    1
K          ds    1
RESTA      ds    1        ; Semiciclos sin adquirir con el ADC
CONTA      ds    1        ; Para un contador ahi...
CONTB      ds    1
```

```

CONTC      ds      1
; PROGRAMA PRINCIPAL
          ORG      FLASH                ; Comienzo del programa
INICIO    BSET    COPD,CONFIG1          ; Deshabilita el COP
          MOV     #$70,RETANTH           ; Valor inicial de retardo
          MOV     #$00,RETANTL           ; Valor inicial de retardo
          JSR     SUBR1                   ; Llama por primera vez
          MOV     #%11000000,CONFIG2     ; SE CONFIGURA EL IRQ
          MOV     #%00000001,INTSCR      ; Tambien configura el IRQ
          MOV     #%00000001,DDRA        ; Configuracion del puerto A
          CLR     PORTA
          MOV     #OUTCOM,TSC1           ; Habilita el Output-Compare TIM1
          CLI
          BCLR   TSTOP,TSC               ; HABILITA LAS IRQ
          BCLR   TSTOP,TSC               ; Se inicia el contador del TIM
ESPERA    NOP
          BRA     ESPERA                  ; SALTA INCONDICIONALMENTE
SUBR1     MOV     #ADCLDR,ADSCR          ; Configura el conversor para tomar UNA
          ; SOLA MUESTRA
          ; del valor del sensor LDR
SAL1      NOP                            ; No realiza nada
          BRCLR  7,ADSCR,SAL1            ; Salta si el bit 7 es cero
          MOV     ADR,SENSOR
          MOV     #ADC_SP,ADSCR          ; Convierte un valor de la referencia
SAL2      NOP                            ; No realiza nada
          BRCLR  7,ADSCR,SAL2            ; Salta si el bit 7 es cero
          MOV     ADR,REF
          LSR     SENSOR                  ; Divide la salida la derecha x2
          LSR     REF                     ; Divide la referencia la derecha x2
          LSR     SENSOR                  ; Divide la salida la derecha x2
          LSR     REF                     ; Divide la referencia la derecha x2
          LSR     SENSOR                  ; Divide la salida la derecha x2
          LSR     REF                     ; Divide la referencia la derecha x2
          LSR     SENSOR                  ; Divide la salida la derecha x2
          LDA     SENSOR
          SUB     REF                     ; Se calcula el error (valor negativo)
          BPL     SAL3
          BMI     SAL4
SAL3      ADD     RETANTL                 ; Parte baja del tiempo de espera
          STA     RETANTL
          LDA     #$00
          BRA     CCCC
SAL4      ADD     RETANTL                 ; Parte baja del tiempo de espera
          STA     RETANTL
          LDA     #$FF
          BRA     CCCC
CCCC      ADC     RETANTH                 ; Parte alta del tiempo de espera
          STA     RETANTH
          SUB     #$95
          BLS     SAL44
          MOV     #$95,RETANTH
          MOV     #$00,RETANTL
SAL44     LDA     RETANTH
          SUB     #$01
          BHS     SAL45
          MOV     #$01,RETANTH
          MOV     #$00,RETANTL

```



```

SAL45 MOV RETANTH,TCH1H
      MOV RETANTL,TCH1L
      RTS

      ; Subrutina que produce el pulso de disparo
DISP BSET PTA0,PORTA ; El pin 7 es PTA0
      MOV #$0F,CONTC ; Antes estaba en FE
SAL6 INC CONTC ; Supuesto retardo
      DEC CONTC
      INC CONTC
      DEC CONTC
      DBNZ CONTC,SAL6 ; Decrementa y salta si no es cero
      BCLR PTA0,PORTA
      RTS

      ; Atencion a la interrupcion del pin IRQ
INTIRQ BSET TRST,TSC ; Se resetea el contador del TIM
        JSR SUBR1
        RTI

INTOC LDA TSC1 ; Lee el estado del CH 1
        BCLR 7,TSC1 ; Borra el flag del timer chanel 1
        JSR DISP
        MOV #$90,CONTA ; Antes estaba en FE
        MOV #$0B,CONTB
SAL66 INC CONTA ; Supuesto retardo
        DEC CONTA
        INC CONTA
        DEC CONTA
        DBNZ CONTA,SAL66 ; Decrementa y salta si no es cero
        MOV #$90,CONTA
        DBNZ CONTB,SAL66 ; Decrementa y salta si no es cero
        BSET PTA0,PORTA ; El pin 7 es PTA0
        JSR DISP
        RTI
ORG RESET
DW INICIO
ORG $FFFA ; Interrupcion del pin IRQ
DW INTIRQ
ORG $FFF4 ; Interrupcion del canal 1 del TIM
DW INTOC
ORG $FFC0
DW $8F89
END

```